



Excel: de functie UNIEK()

Handleiding van Helpmij.nl

Auteur: CorVerm

februari 2022

“ Dé grootste en gratis computerhelpdesk van Nederland ”



We bespreken alweer een prachtige functie in Excel. Ook nu betreft het een **overloop** functie die op een heel gemakkelijke manier gegevens ophaalt uit een bereik of tabel. Niks moeilijks aan. De functie UNIEK() haalt namelijk (duhu) unieke gegevens op, ongesorteerd of gesorteerd.

Opmerking! het bereik of matrix mag geen lege cellen bevatten.

Eerste voorbeeld

	A	B	C	D	E	F
1		VN		Ongesorteerd		
2		Jan		Jan		
3		Cor		Cor		
4		Sjaak		Sjaak		
5		Jeremy		Jeremy		
6		Daan		Daan		
7		Jan		Robin		
8		Cor		Amber		
9		Sjaak				
10		Jeremy				
11		Robin				
12		Jan				
13		Cor				
14		Amber				

Ook nu weer een eenvoudig voorbeeld met wat namen waaruit we unieke namen halen. Met de simpele formule =UNIEK(B2:B14) is dat doel bereikt. Aangezien Excel de volgorde van de namen overneemt zijn die niet gesorteerd. Wil je wel sorteren? Nou dat kan met de formule =SORTEREN(UNIEK(B2:B14)), de namen worden zo van A naar Z gesorteerd. Wil je de selectie omkeren? Ook dat kan. Maar dan moeten er argumenten worden toegevoegd aan de formule.

	A	B	C	D	E	F	G	H
1		VN	Naam	Product				
2		Jan	Verschoor	Geen				
3		Cor	de Hoed	1000 kno...				

Om dat doel te bereiken voer je twee puntkomma's in. De eerste is voor het argument **sorteerindex**. Aangezien we daar geen gebruik van willen maken slaan we dat argument over en typen de tweede puntkomma voor het argument **sorteervolgorde** en kiezen voor **Aflop**. De formule ziet er dan zo uit: =SORTEREN(UNIEK(B2:B14);;-1) en de volgorde is daarmee omgekeerd.

	J	K	L	M
1	2 kolommen			
2	Jan	Verschoor		
3	Cor	den Hoed		
4	Sjaak	Vermolen		
5	Jeremy	Vermeulen		
6	Daan	de Groot		
7	Robin	Vreeze		

De gegevens die je hiernaast ziet zijn

overgenomen uit de afbeelding hieronder.

Door twee kolommen in de formule op te nemen krijg je ook de uitkomst uit beide kolommen. Het bereik is in dit geval **B2:C11**.

Tweede voorbeeld

	A	B	C	D	E	F	G	H
1	VN	Naam	Leeftijd		Ongesorteerd		Gesorteerd	
2	Jan	Verschoor	30		Jan Verschoor		Amber Vermeulen	
3	Cor	den Hoed	32		Cor den Hoed		Cor den Hoed	
4	Sjaak	Vermolen	34		Sjaak Vermolen		Cor Pater	
5	Jeremy	Vermeulen	36		Jeremy Vermeulen		Daan de Groot	
6	Daan	de Groot	38		Daan de Groot		Jan Verschoor	
7	Jan	Verschoor	40		Cor Pater		Jeremy Vermeulen	
8	Cor	Pater	42		Sjaak Vermolen		Robin Vreeze	
9	Sjaak	Vermolen	44		Robin Vreeze		Sjaak Vermolen	
10	Jeremy	Vermeulen	46		Amber Vermeulen		Sjaak Vermolen	
11	Robin	Vreeze	48					
12	Jan	Verschoor	50		=UNIEK(B2:B14&" "&C2:C14)			
13	Cor	Pater	52					
14	Amber	Vermeulen	54					

In dit voorbeeld gebruiken we de functies SORTEREN() en UNIEK(), ook wordt de ampersand (&) gebruikt. In cel F2 staat de formule =UNIEK(B2:B14&" "&C2:C14), hiermee worden de unieke gegevens uit de kolommen B en C gehaald. Door tweemaal een ampersand in het midden van de formule te plaatsen met daartussen twee dubbele quotes (met een spatie er tussen) worden de gegevens samengevoegd. Wil je gegevens gesorteerd weergeven? Gebruik dan weer de functie SORTEREN() in de formule.

Het gebruik van de functie UNIEK() maakt het erg gemakkelijk om gegevens samen te voegen.

Derde voorbeeld

	A	B	C	D	F	G	H
1	VN	Naam	Product		Komt 1 x voor		
2	Jan	Verschoor	Grote beurt		Daan	de Groot	
3	Cor	den Hoed	1000 km beurt		Robin	Vreeze	
4	Sjaak	Vermolen	Oliewissel				
5	Jeremy	Vermeulen	Remblokken				
6	Daan	de Groot	10000 beurt				
7	Jan	Verschoor	APK				
8	Cor	den Hoed	Grote beurt				
9	Sjaak	Vermolen	APK				
10	Jeremy	Vermeulen	APK				
11	Robin	Vreeze	Service beurt				

In dit voorbeeld is het argument [exact_eenmaal] (WAAR of ONWAAR) gebruikt. Het exact_eenmaal argument is een logische waarde die rijen of kolommen retourneert die exact eenmaal in het bereik of de matrix voorkomen. Zoals je aan de uitkomst ziet staan er twee namen in het bereik die exact eenmaal voorkomen.

Zo kun je in een oogopslag zien of een klant maar eenmaal van je diensten gebruik heeft gemaakt en dus in aanmerking komt om eens te benaderen.

De formule in G2 is: =UNIEK(B2:C11;;WAAR). Ook hier zie je tweemaal een puntkomma in de formule, in dit geval is het argument **per_kolom** niet gebruikt.

Een laatste voorbeeld

	VN	Naam		Uniek	Komt 1x voor
2	Jan	Verschoor		7	3
3	Cor	den Hoed			
4	Sjaak	Vermolen	=AANTALARG(UNIEK(B\$2:B\$12))		
5	Jeremy	Vermeulen			
6	Daan	de Groot			
7	Jan	Verschoor			
8	Cor	den Hoed			
9	Sjaak	Vermolen			
10	Jeremy	Vermeulen			
11	Robin	Vreeze			
12	Wim	Vreeze			

Wil je het aantal unieke namen weten? Gebruik dan de functie AANTALARG() in combinatie met de functie UNIEK(). Zoals je in dit voorbeeld ziet telt Excel zeven unieke namen en drie namen die een keer voorkomen.

Er is een maar. Het bereik mag geen lege cellen bevatten. Is dat wel zo dan gebruik je de formule

=AANTALARG(UNIEK(B2:B14))-ALS(AANTAL.LEGE.CELLEN(B2:B14)>0;1;0). Het is maar dat je het weet.

In het voorbeeld hieronder is die formule toegepast in B2.

Voor als je geen Microsoft 365 hebt

	A	B	C	D	E	F	G	H
1	Jan							
2	Cor	5	#DELING DOOR 0!	5				
3	Sjaak	=AANTALARG(UNIEK(A1:A8))-ALS(AANTAL.LEGE.CELLEN(A1:A8)>0;1;0)						
4	Jeremy	=SOMPRODUCT(1/AANTAL.ALS(A1:A8;A1:A8))						
5		=SOMPRODUCT(ALS(AANTAL.ALS(A1:A8;A1:A8)=0;0;1/AANTAL.ALS(A1:A8;A1:A8)))						
6	Jan							
7	Cor							
8	Iman							

Ben je niet in het bezit van Microsoft 365 of Excel 2021 dan kun je de functie SOMPRODUCT() gebruiken. Zoals in het voorbeeld is te zien in C2 krijg je een foutmelding bij het gebruik van de formule =SOMPRODUCT(1/AANTAL.ALS(A1:A8;A1:A8)). Deze formule werkt prima als er geen lege cellen in het bereik zitten. De formule =SOMPRODUCT(ALS(AANTAL.ALS(A1:A8;A1:A8)=0;0;1/AANTAL.ALS(A1:A8;A1:A8))) geeft, hoe dan ook, altijd het juiste resultaat weer.

De functie UNIEK() is, net als de andere Overloop functies, een heel doelgerichte functie. Het bespaart ook nu veel gedoe met matrix formules om de goede uitkomsten te verkrijgen.

Deze functie is toepasbaar in:

Excel voor Microsoft 365, Excel voor Microsoft 365 voor Mac, Webversie van Excel, Excel 2021, Excel 2021 voor Mac, Excel voor iPad, Excel voor iPhone, Excel voor Android-tablets en Excel voor Android-telefoons.

Helpmij.nl